
23 Controller Abstractions

This chapter introduces controller representations for POMDP policies, which
allow policies to maintain their own internal state. These representations can
improve scalability over previous methods that enumerate over belief points.
This chapter presents algorithms that construct controllers using policy iteration,
mathematical programming, and gradient ascent.

23.1 Controllers

A controller is a policy representation that maintains its own internal state. This
controller is represented as a graph consisting of a finite set of nodes X.1 The 1 Such a policy representation is

also called a finite state controller.
Wewill endeavor to refer to the con-
troller states as nodes rather than
states to reduce ambiguitywith the
environment state.

active node changes as actions are taken and new observations are made. Having
a finite set of nodes makes these controllers more computationally tractable than
belief-point methods that must consider the reachable belief space.

Actions are selected according to an action distribution ψ(a | x) that depends
on the current node. When selecting an action, in addition to transitioning to an
unobserved s′ and receiving an observation o, the control state also advances
according to a successor distribution η(x′ | x, a, o). Figure 23.1 shows how these
distributions are used as a controller policy is followed. Algorithm 23.1 provides
an implementation, and example 23.1 shows a controller for the crying baby
problem.

Controllers generalize conditional plans, whichwere introduced in section 20.2.
Conditional plans represent policies as trees, with each node deterministically
assigning an action and each edge specifying a unique successor node. Controllers
represent policies as directed graphs, and actions may have stochastic transitions
to multiple successor nodes. Example 23.2 compares these two representations.

η(x′ | x, a, o) Figure 23.1. In a controller rep-
resentation, the action is sampled
from the action selection distribu-
tion. This action, and the subse-
quent observation it produces, are
used alongside the previous node
x to produce the successor node x′.

We can construct a simple controller for the crying baby problem (ap-
pendix F.7). This policy is shown below as a graph with two nodes, x1 and x2.
When in x1, the controller always ignores the baby. When in x2, the controller
always feeds the baby. If the baby cries, we always transition to x2, and if the
baby is quiet, we always transition to x1.

x1 x2

o = crying

o = quiet

o = quiet o = crying

ψ(ignore | x1) = 1 ψ(feed | x2) = 1

Example 23.1. A two-node exam-
ple controller for the crying baby
problem. This compact represen-
tation captures a straightforward
solution to the crying baby prob-
lem, namely to react immediately
to the most recent observation.

mutable struct ControllerPolicy

𝒫 # problem

X # set of controller nodes

ψ # action selection distribution

η # successor selection distribution

end

function (π::ControllerPolicy)(x)

𝒜, ψ = π.𝒫.𝒜, π.ψ

dist = [ψ[x, a] for a in 𝒜]

return rand(SetCategorical(𝒜, dist))

end

function update(π::ControllerPolicy, x, a, o)

X, η = π.X, π.η

dist = [η[x, a, o, x′] for x′ in X]

return rand(SetCategorical(X, dist))

end

Algorithm 23.1. A finite state con-
troller policy representation for a
POMDP 𝒫. The nodes in X are an ab-
stract representation of reachable
beliefs. Actions and controller suc-
cessor nodes are selected stochasti-
cally. Given a node x, actions are se-
lected following the distribution ψ.
The function π(x) implements this
mechanism to stochastically select
actions. After performing action a

in node x and observing observa-
tion o, the successor is selected fol-
lowing the distribution η. The func-
tion update implements this mech-
anism to stochastically select suc-
cessor nodes.

Consider a three-step conditional plan (below left) compared with the more
general two-node finite state controller (below right) from example 23.1. In
this case, actions and successors are selected deterministically. The deter-
ministic action is marked in the center of a node, and the outgoing edges
represent the deterministic successor nodes. This problem has two actions
(a1 and a2) and two observations (o1 and o2).

a1 a2

o1

o1

o2 o2

o1 = quiet

o2 = crying

a1 = ignore

a2 = feed

The conditional plan performs action a1 first, toggles the previously cho-
sen action if it observes o1, and preserves the previously chosen action if it
observes o2. The controller performs the same logic, with five fewer controller
nodes. Moreover, the controller represents the described infinite horizon pol-
icy perfectly with only two nodes (compared to seven). The conditional plan
cannot capture this infinite horizon policy because it would require a tree of
infinite depth.

Example 23.2. A comparison of a
simple conditional plan with a sim-
ple deterministic controller.

Controllers have several advantages over conditional plans. First, controllers
can provide a more compact representation. The number of nodes in a conditional
plan grows exponentially with depth, but this need not be the case with finite state
controllers. The approximation methods from previous chapters might also not
be as efficient because they must maintain a large set of beliefs and corresponding
alpha vectors. Controllers can be much more compact, considering infinitely
many possible reachable beliefs with a small, finite number of nodes. Another
advantage of controllers is that they do not require that a belief be maintained.2 2 Each controller node corresponds

to a subset of the belief space.
A controller transitions between
these subsets that together cover
the reachable belief space.

The controller itself selects a new node based on each observation, rather than
relying on a belief update, which can be expensive for some domains.

The utility when following a controller can be computed by forming a product
MDP whose state space is X× S . The value of being in state s with node x active
is:

U(x, s) = ∑
a

ψ(a | x)

(
R(s, a) + γ ∑

s′
T(s′ | s, a)∑

o
O(o | a, s′)∑

x′
η(x′ | x, a, o)U(x′, s′)

)
(23.1)

Policy evaluation involves solving the systemof linear equations in equation (23.1).
Alternatively, we can apply iterative policy evaluation as shown in algorithm 23.2.

function utility(π::ControllerPolicy, U, x, s)

𝒮, 𝒜, 𝒪 = π.𝒫.𝒮, π.𝒫.𝒜, π.𝒫.𝒪

T, O, R, γ = π.𝒫.T, π.𝒫.O, π.𝒫.R, π.𝒫.γ

X, ψ, η = π.X, π.ψ, π.η

U′(a,s′,o) = sum(η[x,a,o,x′]*U[x′,s′] for x′ in X)

U′(a,s′) = T(s,a,s′)*sum(O(a,s′,o)*U′(a,s′,o) for o in 𝒪)

U′(a) = R(s,a) + γ*sum(U′(a,s′) for s′ in 𝒮)

return sum(ψ[x,a]*U′(a) for a in 𝒜)

end

function iterative_policy_evaluation(π::ControllerPolicy, k_max)

𝒮, X = π.𝒫.𝒮, π.X

U = Dict((x, s) => 0.0 for x in X, s in 𝒮)

for k in 1:k_max

U = Dict((x, s) => utility(π, U, x, s) for x in X, s in 𝒮)

end

return U

end

Algorithm 23.2. An algorithm for
performing iterative policy evalu-
ation to compute the utility of a fi-
nite state controller π with k_max

iterations. The utility function per-
forms a single step evaluation for
the current controller node x and
state s following equation (23.1).
This algorithm was adapted from
algorithm 7.3, which applies to
MDPs.

If a belief is known, then the current value is

U(x, b) = ∑
s

b(s)U(x, s) (23.2)

We can think of U(x, s) as defining a set of alpha vectors, one for each node x in
X. Each alpha vector αx is defined by αx(s) = U(x, s). The current value for a
given alpha vector is U(x, b) = b>αx.

Given a controller and an initial belief, we can select an initial node by maxi-
mizing:

x∗ = arg max
x

U(x, b) = arg max
x

b>αx (23.3)

23.2 Policy Iteration

Section 20.5 showed how to incrementally add nodes in a conditional plan to
arrive at optimal finite-horizon policy (algorithm 20.8). This section shows how to
incrementally add nodes to a controller to optimize for infinite horizon problems.
Although the policy representation is different, the version of policy iteration for
partially observable problems introduced in this section3 has some similarities 3 The policy iterationmethod given

here was given by E.A. Hansen,
‘‘Solving POMDPs by Searching
in Policy Space,’’ in Conference on
Uncertainty in Artificial Intelligence
(UAI), 1998.

with the policy iteration algorithm for fully observed problems (section 7.4).
Policy iteration (algorithm 23.3) begins with any initial controller, and then

iterates between policy evaluation and policy improvement. In policy evaluation,
we evaluate the utilities U(x, s) by solving equation (23.1). In policy improvement,
we introduce new nodes to our controller. Specifically, we introduce a new node
x′ for every combination of deterministic action assignments ψ(ai | x′) = 1
and deterministic successor selection distributions η(x | x′, a, o). This process
adds |A||X(k)||O| new controller nodes to the set of nodes X(k) at iteration k.4 An

4 Adding all possible combina-
tions is often not feasible. An
alternative algorithm called
bounded policy iteration adds only
one node. P. Poupart and C.
Boutilier, ‘‘Bounded Finite State
Controllers,’’ in Advances in Neural
Information Processing Systems
(NIPS), 2003. Algorithms can
also add a number in between.
Monte Carlo value iteration, for
example, adds O(n|A||X(k)|)
new nodes at each iteration k,
where n is a parameter. H. Bai,
D. Hsu, W. S. Lee, and V.A. Ngo,
‘‘Monte Carlo Value Iteration
for Continuous-State POMDPs,’’
in International Workshop on the
Algorithmic Foundations of Robotics
(WAFR), 2011.

improvement step is demonstrated in example 23.3.
Policy improvement cannot worsen the expected value of the controller policy.

The value of any nodes in X(k) remain unchanged, as they and their reachable
successor nodes remain unchanged. It is guaranteed that if X(k) is not an optimal
controller, then at least one of the new nodes introduced in policy improvement
will have better expected values for some states, and thus the overall controller
must be improved.

Many of the nodes added during policy improvement tend to not improve
the policy. Pruning is conducted after policy evaluation to eliminate unnecessary
nodes. Pruning does not degrade the optimal value function of the controller and
can help reduce the exponential growth in nodes that comes with the improve-
ment step.

struct ControllerPolicyIteration

k_max # number of iterations

eval_max # number of evaluation iterations

end

function solve(M::ControllerPolicyIteration, 𝒫::POMDP)

𝒜, 𝒪, k_max, eval_max = 𝒫.𝒜, 𝒫.𝒪, M.k_max, M.eval_max

X = [1]

ψ = Dict((x, a) => 1.0 / length(𝒜) for x in X, a in 𝒜)

η = Dict((x, a, o, x′) => 1.0 for x in X, a in 𝒜, o in 𝒪, x′ in X)

π = ControllerPolicy(𝒫, X, ψ, η)

for i in 1:k_max

prevX = copy(π.X)

U = iterative_policy_evaluation(π, eval_max)

policy_improvement!(π, U, prevX)

prune!(π, U, prevX)

end

return π

end

function policy_improvement!(π::ControllerPolicy, U, prevX)

𝒮, 𝒜, 𝒪 = π.𝒫.𝒮, π.𝒫.𝒜, π.𝒫.𝒪

X, ψ, η = π.X, π.ψ, π.η

repeatX𝒪 = fill(X, length(𝒪))

assign𝒜X′ = vec(collect(product(𝒜, repeatX𝒪...)))

for ax′ in assign𝒜X′

x, a = maximum(X) + 1, ax′[1]

push!(X, x)

successor(o) = ax′[findfirst(isequal(o), 𝒪) + 1]

U′(o,s′) = U[successor(o), s′]

for s in 𝒮

U[x, s] = lookahead(π.𝒫, U′, s, a)

end

for a′ in 𝒜

ψ[x, a′] = a′ == a ? 1.0 : 0.0

for (o, x′) in product(𝒪, prevX)

η[x, a′, o, x′] = x′ == successor(o) ? 1.0 : 0.0

end

end

end

for (x, a, o, x′) in product(X, 𝒜, 𝒪, X)

if !haskey(η, (x, a, o, x′))

η[x, a, o, x′] = 0.0

end

end

end

Algorithm 23.3. Policy iteration
for a POMDP 𝒫 given a fixed
number of iterations k_max and
number of policy evaluation iter-
ations eval_max. The algorithm it-
eratively applies policy evaluation
from algorithm 23.2 and policy
improvement. Pruning is imple-
mented in algorithm 23.4.

We can apply policy improvement to the crying baby controller from ex-
ample 23.1. The actions are A = {feed, sing, ignore} and observations
are O = {crying,quiet}. The policy improvement backup step results in
|A||X(1)||O| = 3× 22 = 12 new nodes. The new controller policy has nodes
{x1, . . . , x14} and distributions as follows:

Node Action Successors (for all a below)

x3 ψ(feed | x3) = 1 η(x1 | x3, a, crying) = η(x1 | x3, a,quiet) = 1
x4 ψ(feed | x4) = 1 η(x1 | x4, a, crying) = η(x2 | x4, a,quiet) = 1
x5 ψ(feed | x5) = 1 η(x2 | x5, a, crying) = η(x1 | x5, a,quiet) = 1
x6 ψ(feed | x6) = 1 η(x2 | x6, a, crying) = η(x2 | x6, a,quiet) = 1
x7 ψ(sing | x7) = 1 η(x1 | x7, a, crying) = η(x1 | x7, a,quiet) = 1
x8 ψ(sing | x8) = 1 η(x1 | x8, a, crying) = η(x2 | x8, a,quiet) = 1
...

...
...

We have the following controller, with new nodes in blue and the original
two nodes in black:

feed

feed

feed

feed

sing

sing

...

ignore

feed

quiet

crying

cryingquiet

crying
quiet

crying
quiet

crying
quiet

crying
quiet

crying
quiet

crying
quiet

Example 23.3. An example illus-
trating an improvement step as
part of policy iteration on the cry-
ing baby problem with a controller
policy representation.

We prune any new nodes that are identical to existing nodes. We also prune
any new nodes that are dominated by other nodes. A node x is dominated by
another node x′ when:

U(x, s) ≤ U(x′, s) for all s (23.4)

Existing nodes can be pruned as well. Whenever a new node dominates an
existing node, we prune the existing node from the controller. Any transitions
to the deleted node are instead rerouted to the dominating node.5 Example 23.4 5 This process is identical to prun-

ing the new node instead, and up-
dating the dominated node’s ac-
tion and successor links to those
of the new node.

demonstrates evaluation, expansion, and pruning on the crying baby problem.

23.3 Nonlinear Programming

The policy improvement problem can be framed as a single large nonlinear pro-
gramming formulation (algorithm 23.5) that involves simultaneously optimizing
for ψ and η across all nodes.6 This formulation allows general purpose solvers 6 C. Amato, D. S. Bernstein, and

S. Zilberstein, ‘‘Optimizing
Fixed-Size Stochastic Controllers
for POMDPs and Decentralized
POMDPs,’’ Autonomous Agents and
Multi-Agent Systems, vol. 21, no. 3,
pp. 293–320, 2010.

to be applied. The nonlinear programming method directly searches the space
of controllers to maximize the utility of a given initial belief while satisfying
the Bellman equation, equation (23.1). There is no alternating between policy
evaluation and policy improvement steps, and the controller node count remains
fixed.

We use x1 to denote the initial node corresponding to the given initial belief b.
The optimization problem is then:

maximize
U,ψ,η

∑
s

b(s)U(x1, s)

subject to U(x, s) = ∑
a

ψ(a | x)

(
R(s, a) + γ ∑

s′
T(s′ | s, a)∑

o
O(o | a, s′)∑

x′
η(x′ | x, a, o)U(x′, s′)

)
for all x, s

ψ(a | x) ≥ 0 for all x, a

∑
a

ψ(a | x) = 1 for all x

η(x′ | x, a, o) ≥ 0 for all x, a, o, x′

∑
x′

η(x′ | x, a, o) = 1 for all x, a, o

(23.5)

function prune!(π::ControllerPolicy, U, prevX)

𝒮, 𝒜, 𝒪, X, ψ, η = π.𝒫.𝒮, π.𝒫.𝒜, π.𝒫.𝒪, π.X, π.ψ, π.η

newX, removeX = setdiff(X, prevX), []

# prune dominated from previous nodes

dominated(x,x′) = all(U[x,s] ≤ U[x′,s] for s in 𝒮)

for (x,x′) in product(prevX, newX)

if x′ ∉ removeX && dominated(x, x′)

for s in 𝒮

U[x,s] = U[x′,s]

end

for a in 𝒜

ψ[x,a] = ψ[x′,a]

for (o,x′′) in product(𝒪, X)

η[x,a,o,x′′] = η[x′,a,o,x′′]

end

end

push!(removeX, x′)

end

end

# prune identical from previous nodes

identical_action(x,x′) = all(ψ[x,a] ≈ ψ[x′,a] for a in 𝒜)

identical_successor(x,x′) = all(η[x,a,o,x′′] ≈ η[x′,a,o,x′′]

for a in 𝒜, o in 𝒪, x′′ in X)

identical(x,x′) = identical_action(x,x′) && identical_successor(x,x′)

for (x,x′) in product(prevX, newX)

if x′ ∉ removeX && identical(x,x′)

push!(removeX, x′)

end

end

# prune dominated from new nodes

for (x,x′) in product(X, newX)

if x′ ∉ removeX && dominated(x′,x) && x ≠ x′

push!(removeX, x′)

end

end

# update controller

π.X = setdiff(X, removeX)

π.ψ = Dict(k => v for (k,v) in ψ if k[1] ∉ removeX)

π.η = Dict(k => v for (k,v) in η if k[1] ∉ removeX)

end

Algorithm 23.4. The pruning step
of policy iteration. It reduces the
number of nodes in the current
policy π, using the utilities U com-
puted by policy evaluation, and
the previous node list prevX. Its
first step replaces any pointwise
dominated previous nodes by their
improved nodes, marking the re-
dundant node as now dominated.
The second step marks any newly
added nodes that are identical to
previous nodes. The third step
marks any pointwise dominated
new nodes. Finally, all marked
nodes are pruned.

Recall example 23.3. Belowwe show the first iteration of policy iteration using
same initial controller. It consists of the two main steps: policy evaluation
(left) and policy improvement (center), as well as the optional pruning step
(right).

0 0.2 0.4 0.6 0.8 1
−50

−40

−30

−20

−10

P(hungry)

U
(x

,s
)

policy evaluation 1

0 0.2 0.4 0.6 0.8 1
−50

−40

−30

−20

−10

P(hungry)

policy improvement 1

0 0.2 0.4 0.6 0.8 1
−50

−40

−30

−20

−10

P(hungry)

pruning 1

The second iteration of policy iteration follows the same pattern:

The utility has greatly improved after the second iteration to near optimal
values. We see that the prune step removes dominated and duplicate nodes
from previous iterations as well as the current iteration’s new nodes.

Example 23.4. An example of pol-
icy iteration, illustrating the evalu-
ation, improvement, and pruning
steps on the crying baby domain
with a controller policy representa-
tion.

This problem can be written as a quadratically constrained linear program (QCLP),
which can be solved efficiently using a dedicated solver. Example 23.5 demon-
strates this approach.

struct NonlinearProgramming

b # initial belief

ℓ # number of nodes

end

function tensorform(𝒫::POMDP)

𝒮, 𝒜, 𝒪, R, T, O = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.R, 𝒫.T, 𝒫.O

𝒮′ = eachindex(𝒮)

𝒜′ = eachindex(𝒜)

𝒪′ = eachindex(𝒪)

R′ = [R(s,a) for s in 𝒮, a in 𝒜]

T′ = [T(s,a,s′) for s in 𝒮, a in 𝒜, s′ in 𝒮]

O′ = [O(a,s′,o) for a in 𝒜, s′ in 𝒮, o in 𝒪]

return 𝒮′, 𝒜′, 𝒪′, R′, T′, O′

end

function solve(M::NonlinearProgramming, 𝒫::POMDP)

x1, X = 1, collect(1:M.ℓ)

𝒫, γ, b = 𝒫, 𝒫.γ, M.b

𝒮, 𝒜, 𝒪, R, T, O = tensorform(𝒫)

model = Model(Ipopt.Optimizer)

@variable(model, U[X,𝒮])

@variable(model, ψ[X,𝒜] ≥ 0)

@variable(model, η[X,𝒜,𝒪,X] ≥ 0)

@objective(model, Max, b⋅U[x1,:])

@NLconstraint(model, [x=X,s=𝒮],

U[x,s] == (sum(ψ[x,a]*(R[s,a] + γ*sum(T[s,a,s′]*sum(O[a,s′,o]

*sum(η[x,a,o,x′]*U[x′,s′] for x′ in X)

for o in 𝒪) for s′ in 𝒮)) for a in 𝒜)))

@constraint(model, [x=X], sum(ψ[x,:]) == 1)

@constraint(model, [x=X,a=𝒜,o=𝒪], sum(η[x,a,o,:]) == 1)

optimize!(model)

ψ′, η′ = value.(ψ), value.(η)

return ControllerPolicy(𝒫, X,

Dict((x, 𝒫.𝒜[a]) => ψ′[x, a] for x in X, a in 𝒜),

Dict((x, 𝒫.𝒜[a], 𝒫.𝒪[o], x′) => η′[x, a, o, x′]

for x in X, a in 𝒜, o in 𝒪, x′ in X))

end

Algorithm 23.5. A nonlinear pro-
gramming approach to compute
the optimal fixed-size controller
policy for POMDP 𝒫 starting at ini-
tial belief b. The size of the finite
state controller is specified by num-
ber of nodes ℓ.

Below are optimal fixed-size controllers computed using nonlinear program-
ming for the crying baby problem with b0 = [0.5, 0.5]. The top node is x1.

With k = 1, the optimal policy is to simply ignore forever. With k = 2, the
optimal policy is to ignore until crying is observed, at which point the best
action is to feed the child, and then return to ignoring. This policy is close to
optimal for the infinite horizon crying baby POMDP. With k = 3, the optimal
policy essentially remains unchanged from when k = 2.

Example 23.5. An example of the
nonlinear programming algorithm
for controllers with a fixed size of
k set to 1, 2, and 3. Each row shows
the policy and its corresponding
utilities (alpha vectors) on the left
and right, respectively. The stochas-
tic controllers are shown as circles,
with the most likely action in the
middle. The outgoing edges show
successor node selections given an
observation. The stochasticity in
node actions and successors are
shown as opacity (more opaque is
higher probability, more transpar-
ent is lower probability).

23.4 Gradient Ascent

A fixed-size controller policy can be iteratively improved using gradient ascent
(covered in appendix A.11).7 Though the gradient is challenging to compute, 7 N. Meuleau, K.-E. Kim, L. P. Kael-

bling, and A.R. Cassandra, ‘‘Solv-
ing POMDPs by Searching the
Space of Finite Policies,’’ in Confer-
ence on Uncertainty in Artificial Intel-
ligence (UAI), 1999.

this opens up controller optimization to a wide variety of gradient-based opti-
mization techniques. Algorithm 23.6 implements controller gradient ascent using
algorithm 23.7.

Consider an explicit description of the nonlinear problem from section 23.3.
For initial belief b and an arbitrary initial controller node x1, we seek to maximize:

∑
s

b(s)U(x1, s) (23.6)

with the utility U(x, s) defined by the Bellman equation for all x and s:

U(x, s) = ∑
a

ψ(a | x)

(
R(s, a) + γ ∑

s′
T(s′ | s, a)∑

o
O(o | a, s′)∑

x′
η(x′ | x, a, o)U(x′, s′)

)
(23.7)

In addition, ψ and η must also be proper probability distributions. To apply
gradient ascent, it is more convenient to rewrite this problem using linear algebra.

We define the transition function with a controller, which has a state space
X × S . For any fixed-size controller policy parameterized by θ = (ψ, η), the
transition matrix Tθ ∈ R|X×S|×|X×S| is:

Tθ((x, s), (x′, s′)) = ∑
a

ψ(x, a)T(s, a, s′)∑
o

O(a, s′, o)η(x, a, o, x′) (23.8)

The reward for a parameteried policy is represented as a vector rθ ∈ R|X×S|:

rθ((x, s)) = ∑
a

ψ(x, a)R(s, a) (23.9)

The Bellman equation for utility uθ ∈ R|X×S| is then:

uθ = rθ + γTθuθ (23.10)

We can consider an initial node-belief vector β ∈ R|X×S| with βxs = b(s) if
x = x1 and βxs = 0 otherwise. A utility vector uθ ∈ R|X×S| also is defined over
the nodes X and states S for any of these fixed-size parameterized controller
policies θ = (ψ, η). We now seek to maximize:

β>uθ (23.11)

We begin by rewriting equation (23.10):

uθ = rθ + γTθuθ (23.12)
(I− γTθ)uθ = rθ (23.13)

uθ = (I− γTθ)−1rθ (23.14)
uθ = Z−1rθ (23.15)

with Z = I− γTθ for convenience. In order to perform gradient ascent, we need
to know the partial derivatives of equation (23.15) with respect to the policy
parameters:

∂uθ
∂θ

=
∂Z−1

∂θ
rθ + Z−1 ∂rθ

∂θ
(23.16)

= −Z−1 ∂Z
∂θ

Z−1rθ + Z−1 ∂rθ
∂θ

(23.17)

= Z−1
(

∂rθ
∂θ
− ∂Z

∂θ
Z−1rθ

)
(23.18)

with ∂θ referring to both ∂ψ(x̂, â) and ∂η(x̂, â, ô, x̂′) for convenience.
Computing the partial derivatives of Z and rθ results in four equations:

∂rθ((x, s))
∂ψ(x̂, â)

=

{
R(s, a) if x = x̂
0 otherwise

(23.19)

∂rθ((x, s))
∂η(x̂, â, ô, x̂′)

= 0 (23.20)

∂Z((x, s), (x′, s′))
∂ψ(x̂, â)

=

{
−γT(s, â, s′)∑o O(â, s′, o)η(x̂, â, o, x′) if x = x̂
0 otherwise

(23.21)

∂Z((x, s), (x′, s′))
∂η(x̂, â, ô, x̂′)

=

{
−γψ(x̂, â)T(s, â, s′)O(â, s′, ô)η(x̂, â, ô, x′) if x = x̂ and x′ = x̂′

0 otherwise
(23.22)

Finally, these four gradients are used in the utility gradients from equation (23.18):

∂uθ
∂ψ(x̂, â)

= Z−1
(

∂rθ
∂ψ(x̂, â)

− ∂Z
∂ψ(x̂, â)

Z−1rθ

)
(23.23)

∂uθ
∂η(x̂, â, ô, x̂′)

= Z−1
(

∂rθ
∂η(x̂, â, ô, x̂′)

− ∂Z
∂η(x̂, â, ô, x̂′)

Z−1rθ

)
(23.24)

We finally can return to the original objective in equation (23.11). Controller
gradient ascent starts with a fixed number of nodes in X and an arbitrary policy
ψ and η. At iteration k, it updates these parameters as follows:

ψk+1(x, a) = ψk(x, a) + αβ>
∂u
θk

∂ψk(x̂, â)
(23.25)

ηk+1(x, a, o, x′) = ηk(x, a, o, x′) + αβ>
∂u
θk

∂ηk(x̂, â, ô, x̂′)
(23.26)

with gradient step size α > 0. The distributions ψk+1 and ηk+1 may no longer be
valid after this update. They may need non-negativity to be enforced and to be
normalized.8 Example 23.6 demonstrates this process. 8 This procedure is more generally

called projecting to the probability
simplex.

The optimization objective in equation (23.6) is not necessarily convex.9 Hence,
9 This objective is distinct from the
utility U(x, b) = ∑s b(s)U(x, s),
which is guaranteed to be piece-
wise linear and convex with re-
spect to the belief state b as dis-
cussed in section 20.3.

normal gradient ascent can converge to a local optimum depending on the initial
controller. Adaptive gradient algorithms can be applied to help smooth and speed
convergence.

23.5 Summary

• Controllers are a policy representation that do not rely on exploring or main-
taining beliefs.

• Controllers consist of nodes, an action selection function, and a successor
selection function.

• Nodes and the controller graph are abstract, however, they can be interpreted
as sets of the countably infinite reachable beliefs; the value function for a node
can also be interpreted as an alpha vector.

• Policy iteration iterates between policy evaluation that computes the utilities
for each node, and policy improvement that adds new nodes, with the option
to introduce pruning.

• Nonlinear programming reformulates finding the optimal fixed-sized con-
troller as a general optimization problem , allowing for off-the-shelf solvers
and techniques to be used.

• Controller gradient ascent climbs in the space of policies to improve the value
function directly, benefiting from an explicit POMDP-based gradient step.

struct ControllerGradient

b # initial belief

ℓ # number of nodes

α # gradient step

k_max # maximum iterations

end

function solve(M::ControllerGradient, 𝒫::POMDP)

𝒜, 𝒪, ℓ, k_max = 𝒫.𝒜, 𝒫.𝒪, M.ℓ, M.k_max

X = collect(1:ℓ)

ψ = Dict((x, a) => rand() for x in X, a in 𝒜)

η = Dict((x, a, o, x′) => rand() for x in X, a in 𝒜, o in 𝒪, x′ in X)

π = ControllerPolicy(𝒫, X, ψ, η)

for i in 1:k_max

improve!(π, M, 𝒫)

end

return π

end

function improve!(π::ControllerPolicy, M::ControllerGradient, 𝒫::POMDP)

𝒮, 𝒜, 𝒪, X, x1, ψ, η = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, π.X, 1, π.ψ, π.η

n, m, z, b, ℓ, α = length(𝒮), length(𝒜), length(𝒪), M.b, M.ℓ, M.α

∂U′∂ψ, ∂U′∂η = gradient(π, M, 𝒫)

UIndex(x, s) = (s - 1) * ℓ + (x - 1) + 1

E(U, x1, b) = sum(b[s]*U[UIndex(x1,s)] for s in 1:n)

ψ′ = Dict((x, a) => 0.0 for x in X, a in 𝒜)

η′ = Dict((x, a, o, x′) => 0.0 for x in X, a in 𝒜, o in 𝒪, x′ in X)

for x in X

ψ′x = [ψ[x, a] + α * E(∂U′∂ψ(x, a), x1, b) for a in 𝒜]

ψ′x = project_to_simplex(ψ′x)

for (aIndex, a) in enumerate(𝒜)

ψ′[x, a] = ψ′x[aIndex]

end

for (a, o) in product(𝒜, 𝒪)

η′x = [(η[x, a, o, x′] +

α * E(∂U′∂η(x, a, o, x′), x1, b)) for x′ in X]

η′x = project_to_simplex(η′x)

for (x′Index, x′) in enumerate(X)

η′[x, a, o, x′] = η′x[x′Index]

end

end

end

π.ψ, π.η = ψ′, η′

end

function project_to_simplex(y)

u = sort(copy(y), rev=true)

i = maximum([j for j in eachindex(u)

if u[j] + (1 - sum(u[1:j])) / j > 0.0])

δ = (1 - sum(u[j] for j = 1:i)) / i

return [max(y[j] + δ, 0.0) for j in eachindex(u)]

end

Algorithm 23.6. An implementa-
tion of a controller gradient ascent
algorithm for POMDP 𝒫 at initial
belief b. The controller itself has a
fixed-size of ℓ nodes. It is improved
over k_max iterations by following
the gradient of the controller, with
a step size of α, to maximally im-
prove the value of the initial belief.

2021-01-07 22:28:48-08:00, revision abe81c1, draft: send comments to bugs@algorithmsbook.com



23.6. exercises 481

function gradient(π::ControllerPolicy, M::ControllerGradient, 𝒫::POMDP)

𝒮, 𝒜, 𝒪, T, O, R, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.T, 𝒫.O, 𝒫.R, 𝒫.γ

X, x1, ψ, η = π.X, 1, π.ψ, π.η

n, m, z = length(𝒮), length(𝒜), length(𝒪)

X𝒮 = vec(collect(product(X, 𝒮)))

T′ = [sum(ψ[x, a] * T(s, a, s′) * sum(O(a, s′, o) * η[x, a, o, x′]

for o in 𝒪) for a in 𝒜) for (x, s) in X𝒮, (x′, s′) in X𝒮]

R′ = [sum(ψ[x, a] * R(s, a) for a in 𝒜) for (x, s) in X𝒮]

Z = 1.0I(length(X𝒮)) - γ * T′

invZ = inv(Z)

∂Z∂ψ(hx, ha) = [x == hx ? (-γ * T(s, ha, s′)

* sum(O(ha, s′, o) * η[hx, ha, o, x′]

for o in 𝒪)) : 0.0

for (x, s) in X𝒮, (x′, s′) in X𝒮]

∂Z∂η(hx, ha, ho, hx′) = [x == hx && x′ == hx′ ? (-γ * ψ[hx, ha]

* T(s, ha, s′) * O(ha, s′, ho)) : 0.0

for (x, s) in X𝒮, (x′, s′) in X𝒮]

∂R′∂ψ(hx, ha) = [x == hx ? R(s, ha) : 0.0 for (x, s) in X𝒮]

∂R′∂η(hx, ha, ho, hx′) = [0.0 for (x, s) in X𝒮]

∂U′∂ψ(hx, ha) = invZ * (∂R′∂ψ(hx, ha) - ∂Z∂ψ(hx, ha) * invZ * R′)

∂U′∂η(hx, ha, ho, hx′) = invZ * (∂R′∂η(hx, ha, ho, hx′)

- ∂Z∂η(hx, ha, ho, hx′) * invZ * R′)

return ∂U′∂ψ, ∂U′∂η

end

Algorithm 23.7. The gradient step
of the controller gradient ascent.
It constructs the gradients of the
utility U with respect to the policy
∂U′∂ψ and ∂U′∂η.

23.6 Exercises

Exercise 23.1. List any advantages that a controller policy representation has over condi-
tional plan and belief-based representations.

Solution: Unlike tree-based conditional plans, controllers can represent policies that can be
executed indefinitely. They do not have to grow exponentially in size with the horizon.

Compared to belief-based representations, the number of parameters in a controller
representation tends to be far less than the number of alpha vectors for larger problems.
We can also more easily optimize controllers for a fixed amount of memory.

Controllers will never divide by zero the way that belief-based policies can during
execution. Belief-based methods require maintaining a belief. The discrete state filter from
equation (19.7) will divide by zero if an impossible observation is made. This can happen
when a noisy observation from a sensor returns an observation that the models of T(s, a, s′)
and O(o | a, s′) does not accurately capture.

Exercise 23.2. Controller policy iteration only adds nodes with deterministic action selec-
tion functions and successor distributions. Does this mean that the resulting controller is
necessarily suboptimal?

Consider the catch problem (appendix F.9) with a uniform initial belief b1.
The figures below show the utility of the policy over gradient ascent iteration
applied to the catch problem with k = 3 nodes. The left node is x1.

60 10 10

catch, drop

catch, drop

catch, drop
drop

catch, drop

drop
catch, drop

catch, drop

At iteration 1, the policy is essentially random, both in action selection
and successor selection.

50 50 50

catch, drop

catch, drop
catch, drop

catch, drop
drop

catch, drop

catch, drop
catch, drop

catch, drop

At iteration 50, the agent has determined a reasonable distance to throw
the ball (50) but still has not used its three nodes to remember anything
useful.

40 50 50

catch, drop

catch

drop
catch

catch

drop
catch, drop

At iteration 500, the policy has constructed a reasonable plan, given its
fixed three nodes of memory. It first tries throwing the ball at a distance of 40.
If the child catches the ball, then it increases the range to 50. It uses the final
node to remember how many times the child caught the ball (up to twice)
to choose the distance.

Example 23.6. An example of the
controller gradient algorithm for
controllers with a fixed size of ` =
3. The policy is shown to refine
itself over the algorithm’s itera-
tions. The agent incrementally de-
termines how to best use its fixed
number of nodes, resulting in a rea-
sonable and interpretable policy
upon convergence.

Solution: Controller policy iteration is guaranteed to converge on an optimal policy in
the limit. However, the method cannot find more compact representations of optimal
controller policies that may require stochastic nodes.

Exercise 23.3. Prove that in policy iteration, any pruned node does not affect the utility.

Solution: Let x′ be the new node from some iteration i, and x be a previous node from
iteration i− 1.

By construction, η(x′, a, o, x) defines all new nodes x′ to only have a successor x from
the previous iteration. Thus for each state s, U(i)(x′, s) only sums over the successors
U(i−1)(x, s′) in equation (23.1). This means that the other utilities in iteration i, including
a self-loop to x itself, do not affect the utility U(i)(x′, s). Since the initial node is chosen
by equation (23.3), we must ensure the utility with and without the pruned node at all
beliefs is the same. A node is pruned in one of two ways.

First, x′ obtains a higher utility over all states than its pruned successor x. Formally,
U(i)(x, s) ≤ U(i−1)(x′, s) for all s. The prune step replaces x with x′, including U, ψ, and
η. By construction, U has not decreased at any state s.

Second, x is identical to an existing previous node x′. Note that this means the transition
η(x, a, o, x′) = η(x′, a, o, x′). This means that the utility is identical except that x is reduced
by γ; in other words, γU(i)(x, s) = U(i−1)(x, s) by equation (23.1). It does not affect the
final utility to prune x.

Exercise 23.4. Devise an algorithm that uses the nonlinear program algorithm to find the
minimum fixed-sized controller required to obtain the optimality of a large fixed-sized
controller of size `. You can assume the nonlinear optimizer returns the optimal policy in
this case.

Solution: The idea is to create an outer loop that increments the fixed-size of the controller,
after knowing the utility of the large fixed-sized controller. First, we must compute the
large fixed-sized controller’s utility U∗ = ∑s b1(s)U(x1, s) at initial node x1 and initial
belief b1. Next, we create a loop that increments the size ` of the controller. At each step, we
evaluate the policy and compute the utility U`. By our assumption, the controller returned
produces a globally optimal utility for the fixed size `. Once we arrive at a utility U`, if we
see that U` = U∗, then we stop and return the policy.

Exercise 23.5. Analyze the controller gradient ascent algorithm’s gradient step. Assume
that |S| is larger than |A| and |O|. What is the most computationally expensive part of the
gradient step? How might this be improved?

Solution: Computing the inverse Z−1 = (I− γTθ) is the most computationally expensive
part of the gradient step, as well as the entire gradient algorithm. The matrix Z is of size
|X× S|. Gauss–Jordan elimination requires O(|X× S|3) operations, though the 3 in the
exponent can be reduced to 2.3728639 using a state-of-the-art matrix inversion algorithm.10

10 F. L. Gall, ‘‘Powers of Tensors and
Fast Matrix Multiplication,’’ in In-
ternational Symposium on Symbolic
and Algebraic Computation (ISSAC),
2014.

The creation of the temporary matrix Tθ also requires O(|X× S|2|A ×O|) operations to

support computing the inverse. All other loops and other temporary array creations require
far fewer operations. This can be improved using an approximate inverse technique.

